

Strictly Monotone Brouwer Trees for Well-founded Recursion Over Multiple Arguments

Joseph Eremondi

Laboratory for the Foundations of Computer Science
University of Edinburgh
Scotland, United Kingdom
joe.eremondi@ed.ac.uk

Abstract

Ordinals can be used to prove the termination of dependently typed programs. Brouwer trees are a particular ordinal notation that make it very easy to assign sizes to higher order data structures. They extend unary natural numbers with a limit constructor, so a function's size can be the least upper bound of the sizes of values from its image. These can then be used to define well-founded recursion: any recursive calls are allowed so long as they are on values whose sizes are strictly smaller than the current size.

Unfortunately, Brouwer trees are not algebraically well-behaved. They can be characterized equationally as a join-semilattice, where the join takes the maximum of two trees. However, this join does not interact well with the successor constructor, so it does not interact properly with the strict ordering used in well-founded recursion.

We present Strictly Monotone Brouwer trees (SMB-trees), a refinement of Brouwer trees that are algebraically well-behaved. SMB-trees are built using functions with the same signatures as Brouwer tree constructors, and they satisfy all Brouwer tree inequalities. However, their join operator distributes over the successor, making them suited for well-founded recursion or equational reasoning.

This paper teaches how, using dependent pairs and careful definitions, an ill-behaved definition can be turned into a well-behaved one. Our approach is axiomatically lightweight: it does not rely on Axiom K, univalence, quotient types, or Higher Inductive Types. We implement a recursively-defined maximum operator for Brouwer trees that matches on successors and handles them specifically. Then, we define SMB-trees as the subset of Brouwer trees for which the recursive

maximum computes a least upper bound. Finally, we show that every Brouwer tree can be transformed into a corresponding SMB-tree by infinitely joining it with itself. All definitions and theorems are implemented in Agda.

CCS Concepts • Theory of computation → Type theory; • Software and its engineering → Software verification;

Keywords dependent types, Brouwer trees, well founded recursion

ACM Reference Format:

Joseph Eremondi. 2024. Strictly Monotone Brouwer Trees for Well-founded Recursion Over Multiple Arguments. In *Proceedings of the 13th ACM SIGPLAN International Conference on Certified Programs and Proofs (CPP '24)*, January 15–16, 2024, London, UK. ACM, New York, NY, USA, 13 pages. <https://doi.org/10.1145/3636501.3636948>

1 Introduction

1.1 Recursion and Dependent Types

Dependently typed programming languages bridge the gap between theorem proving and programming. In languages like Agda [Norell 2009], Coq [Bertot and Castéran 2004], Idris [Brady 2021], and Lean [de Moura et al. 2015], one can write programs, specifications, and proofs that programs meet those specifications, all using a unified language.

One challenge in writing dependently typed code is proving termination. Functions in dependently typed languages are typically required to be *total*: they must provably halt in all inputs. This is necessary both to ensure that type checking terminates and to prevent false results from being accidentally proven. Since the halting problem is undecidable, recursively-defined functions must be written in such a way that the type checker can mechanically deduce termination. Some functions only make recursive calls to structurally-smaller arguments, so their termination is apparent to the compiler. However, some functions are not easily expressed using structural recursion. For such functions, the programmer must instead use *well-founded recursion*, showing that there is some ordering, with no infinitely-descending chains, for which each recursive call is strictly smaller according to this ordering. For example, a typical quicksort is not structurally recursive, but can use well-founded recursion on the length of the lists being sorted.

Permission to make digital or hard copies of all or part of this work for personal or classroom use is granted without fee provided that copies are not made or distributed for profit or commercial advantage and that copies bear this notice and the full citation on the first page. Copyrights for components of this work owned by others than the author(s) must be honored. Abstracting with credit is permitted. To copy otherwise, or republish, to post on servers or to redistribute to lists, requires prior specific permission and/or a fee. Request permissions from permissions@acm.org.

CPP '24, January 15–16, 2024, London, UK

© 2024 Copyright held by the owner/author(s). Publication rights licensed to ACM.

ACM ISBN 979-8-4007-0488-8/24/01...\$15.00

<https://doi.org/10.1145/3636501.3636948>

1.2 Ordinals

While numeric orderings work for first-order data, they are ill-suited to recursion over higher-order data structures, where some fields contain functions. Instead, one must use *ordinals* to assign a size to such data structures, so that even when a structure represents infinite data, only a finite number of recursive calls are made when traversing it. In classical mathematics, ordinals are totally ordered and straightforward to reason about. They have many different representations, all of which are equivalent. However, in constructive theories, such as those underlying dependently typed languages, there are many representations of ordinals which are not equivalent. Different constructive ordinal notations have different capabilities, each with their own advantages and disadvantages.

1.3 Contributions

This work defines *strictly monotone Brouwer Trees*, henceforth SMB-trees, a new presentation of ordinals that hit a sort of sweet-spot for defining functions by well-founded recursion. Specifically, SMB-trees:

- Are strictly ordered by a well-founded relation;
- Have a maximum operator which computes a least-upper bound;
- Are *strictly-monotone* with respect to the maximum: if $a < b$ and $c < d$, then $\max a c < \max b d$;
- Can compute the limits of arbitrary sequences;
- Are light in axiomatic requirements: they are defined without using axiom K, univalence, quotient types, or higher inductive types.

The novel insight behind our contribution is that there is a subset of Brouwer trees which behave in the way we want. Specifically, the ability of Brouwer trees to take the limit of a sequence allows us to apply operations to an ordinal an infinite number of times, exposing properties that do not hold for finite applications but do hold in the limit.

1.4 Uses for SMB-trees

Well-founded Recursion Having a maximum operator for ordinals is particularly useful when traversing over multiple higher order data structures in parallel, where neither argument takes priority over the other. In such a case, a lexicographic ordering cannot be used.

As an example, consider a unification algorithm that merges two higher order data structures, such as a unifier for a strongly typed encoding of dependent types, and suppose that α -renaming or some other restriction prevents structural recursion from being used.

To solve a unification problem $\Sigma(x : A). B = \Sigma(x : C). D$ we must recursively solve $A = C$ and $\forall x. B[x] = D[x]$. However, the types of the variables in the latter equation are different. So after computing the unification of A and C , we may need to traverse B and D and convert terms from type

A or C to their unification. If such a conversion is defined mutually with unification, then it must work on a pair of types strictly smaller than $\Sigma(x : A). B, \Sigma(x : C). D$.

To assign sizes to such a procedure, we need a few features. First, we need a maximum operator, so that we can bound the size of unifying A and C by their maximum size. Second, the operator should be strictly monotone, so that the recursive call unifying A and C is on a strictly smaller size. Third, the maximum should be commutative: we need the size of the nested pairs $((A, B), (C, D))$ to be the same as $((A, C), (B, D))$, so that a recursive call on arguments whose size is bounded by the maximum of (A, C) will still be strictly smaller than the initial size of $((A, B), (C, D))$. One such call would be the procedure converting from type A to the solution of $A = C$. Lexicographic orderings lack this commutativity, and are too restrictive for situations such as this.

This style of induction was used to prove termination in a syntactic model of gradual dependent types [Eremondi 2023b]. There, Brouwer trees were used to establish termination of recursive procedures that combined the type information in two imprecise types. The decreasing metric was the maximum size of the codes for the types being combined. Brouwer trees' arbitrary limits were used to assign sizes to dependent function and product types, and the strict monotonicity of the maximum operator was essential for proving that recursive calls were on strictly smaller arguments.

We want to enable the programmer to specify complex relationships between the sizes of multiple arguments and to deduce facts about those sizes in a principled way.

Syntactic Models and Sized Types An alternate view of our contribution is as a tool for modelling sized types [Hughes et al. 1996]. The implementation of sized types in Agda has been shown to be unsound [Agda-Developers 2017], due to the interaction between propositional equality and the top size ∞ satisfying $\infty < \infty$. Chan [2022] defines a dependently typed language with sized types that does not have a top size, proving it consistent using a syntactic model based on Brouwer trees.

SMB-trees provide the capability to extend existing syntactic models to sized types with a maximum operator. This brings the capability of consistent sized types closer to feature parity with Agda, which has a maximum operator for its sizes, while still maintaining logical consistency.

Algebraic Reasoning Another advantage of SMB-trees is that they allow Brouwer trees to be understood using algebraic tools. In algebraic terminology, SMB-trees satisfy the following algebraic laws, up to the equivalence relation defined by $s \approx t := s \leq t \leq s$

- Join-semilattice: the binary \max is associative, commutative, and idempotent;
- Bounded: there is a least tree Z such that $\max t Z \approx t$;

- Inflationary endomorphism: there is a successor operator \uparrow such that $\max(\uparrow t) t \approx \uparrow t$ and $\uparrow(\max s t) \approx \max(\uparrow s)(\uparrow t)$;

Bezem and Coquand [2022] describe a polynomial time algorithm for solving equations in such an algebra, and describe its usefulness for solving constraints involving universe levels in dependent type checking. While equations involving limits of infinite sequences are undecidable, the inflationary laws could be used to automatically discharge some equations involving sizes. This algebraic presentation is particularly amenable to solving equations using free extensions of algebras [Allais et al. 2023; Corbyn 2021].

1.5 Implementation

We have implemented SMB-trees in Agda 2.6.4 with std-lib 1.7.3. Our library specifically avoids Agda-specific features such as cubical type theory or Axiom K, so we expect that it can be easily ported to other proof assistants.

This paper is written as a literate Agda document, and the definitions given in the paper are valid Agda code. For several definitions, only the type is presented, with the body omitted due to space restrictions. The full implementation is open source [Eremondi 2023a].

2 Brouwer Trees: An Introduction

Brouwer trees [Church 1938; Kleene 1938] are a simple but elegant tool for proving termination of higher-order procedures. Traditionally, they are defined as follows:

```
data SmallTree : Set where
  Z : SmallTree
  ↑ : SmallTree → SmallTree
  Lim : (N → SmallTree) → SmallTree
```

A Brouwer tree is either zero, the successor of another Brouwer tree, or the limit of a countable sequence of Brouwer trees. However, these are quite weak, in that they can only take the limit of countable sequences. To represent the limits of uncountable sequences, we can parameterize our definition over some Universe à la Tarski:

```
module Brouwer {ℓ}
  (C : Set ℓ)
  (El : C → Set ℓ)
  (CN : C) (CNIso : Iso (El CN) N) where
```

Our module is parameterized over a universe level, a type \mathbb{C} of codes, and an “elements-of” interpretation function El , which computes the type represented by each code. We require a code whose interpretation is isomorphic to the natural numbers, as this is essential to our construction in Section 4.1. This also ensures that our trees are at least as powerful as `SmallTree`. Increasingly larger ordinals can be obtained by setting $\mathbb{C} := \text{Set } \ell$ and $El := \text{id}$ for increasing ℓ . However, by defining an inductive-recursive universe, one

can still capture limits over some non-countable types, since `Tree` is in `Set 0` whenever \mathbb{C} is.

Given our universe of codes, we generalize limits to any function whose domain is the interpretation of some code.

```
data Tree : Set ℓ where
  Z : Tree
  ↑ : Tree → Tree
  Lim : (c : C) → (f : El c → Tree) → Tree
```

The small limit constructor can be recovered from the natural-number code

```
NLim : (N → Tree) → Tree
NLim f = Lim CN (λ cn → f (Iso.fun CNIso cn))
```

Brouwer trees are the quintessential example of a higher-order inductive type¹: each tree is built using smaller trees or functions producing smaller trees, which is essentially a way of storing a possibly infinite number of smaller trees.

2.1 Ordering Trees

Our ultimate goal is to have a well-founded ordering², so we define a relation to order Brouwer trees.

```
data _≤_ : Tree → Tree → Set ℓ where
  ≤-Z : ∀ {t} → Z ≤ t
  ≤-sucMono : ∀ {t1 t2}
    → t1 ≤ t2
    → ↑ t1 ≤ ↑ t2
  ≤-cocone : ∀ {t} {c : C} (f : El c → Tree) (k : El c)
    → t ≤ f k
    → t ≤ Lim c f
  ≤-limiting : ∀ {t} {c : C}
    → (f : El c → Tree)
    → (∀ k → f k ≤ t)
    → Lim c f ≤ t
```

There are four constructors. First, zero is less than any other tree. Second, the successor operator is monotone: if $t_1 \leq t_2$, then $\uparrow t_1 \leq \uparrow t_2$. Finally, there are two constructors which establish that $\text{Lim } c f$ denotes the least upper bound of the image of f . First `≤-cocone` establishes that $f x \leq \text{Lim } c f$, i.e., it is an upper bound on the image of f . Second, `≤-limiting` establishes that if a value is an upper bound on the image of f , then $\text{Lim } c f$ is less than that value, i.e. it is the least of all upper bounds. The constructor names and types are adapted from Kraus et al. [2023], although we change the definition of `≤-cocone` slightly so that we do not need a separate constructor for transitivity.

¹Not to be confused with Higher Inductive Types (HITs) from Homotopy Type Theory [Univalent Foundations Program 2013]

²Technically, this is a well-founded quasi-ordering because there are pairs of trees which are related by both \leq and \geq , but which are not propositionally equal.

This relation is reflexive:

```

≤-refl : ∀ t → t ≤ t
≤-refl Z = ≤-Z
≤-refl (↑ t) = ≤-sucMono (≤-refl t)
≤-refl (Lim c f)
  = ≤-limiting f (λ k → ≤-cocone f k (≤-refl (f k)))

```

Crucially, it is also transitive, making the relation a preorder.

```

≤-trans : ∀ {t1 t2 t3} → t1 ≤ t2 → t2 ≤ t3 → t1 ≤ t3
≤-trans ≤-Z p23 = ≤-Z
≤-trans (≤-sucMono p12) (≤-sucMono p23)
  = ≤-sucMono (≤-trans p12 p23)
≤-trans p12 (≤-cocone f k p23)
  = ≤-cocone f k (≤-trans p12 p23)
≤-trans (≤-limiting f x) p23
  = ≤-limiting f (λ k → ≤-trans (x k) p23)
≤-trans (≤-cocone f k p12) (≤-limiting f x)
  = ≤-trans p12 (x k)

```

We create an infix version of transitivity for more readable construction of proofs:

```

_≤_§_ : ∀ {t1 t2 t3} → t1 ≤ t2 → t2 ≤ t3 → t1 ≤ t3
lt1 ≤_§_ lt2 = ≤-trans lt1 lt2

```

A useful property is that limits of sequences are related if the sequences are related element-wise:

```

extLim : ∀ {c : C}
  → (f1 f2 : El c → Tree)
  → (∀ k → f1 k ≤ f2 k)
  → Lim c f1 ≤ Lim c f2
extLim {c = c} f1 f2 all
  = ≤-limiting f1 (λ k → ≤-cocone f2 k (all k))

```

Strict Ordering We can define a strictly-less-than relation in terms of our less-than relation and the successor constructor:

```

_<_ : Tree → Tree → Set ℓ
t1 < t2 = ↑ t1 ≤ t2

```

That is, t_1 is strictly smaller than t_2 if the tree one-size larger than t_1 is as small as t_2 . The fact that $\uparrow t$ is always strictly larger than t is a key property of ordinals. Adding one element to a countably-infinite set does not change its cardinality, but taking the successor of an infinite ordinal produces something larger, which is why they are useful for assigning sizes to infinite data.

This relation has the properties one expects of a strictly-less-than relation: it is a transitive sub-relation of the less-than relation, every tree is strictly less than its successor, and no tree is strictly smaller than zero.

```

≤↑t : ∀ t → t ≤ ↑t
≤↑t Z = ≤-Z

```

```

≤↑t (↑ t) = ≤-sucMono (≤↑t t)
≤↑t (Lim c f)
  = ≤-limiting f λ k →
    (≤↑t (f k))
    ≤_§_ (≤-sucMono (≤-cocone f k (≤-refl (f k))))

```

```

<-in-≤ : ∀ {x y} → x < y → x ≤ y
<-≤-in-< : ∀ {x y z} → x < y → y ≤ z → x < z
≤-<-in-< : ∀ {x y z} → x ≤ y → y < z → x < z
¬<Z : ∀ t → ¬(t < Z)

```

2.2 Well-founded Induction

Here we recall the definition of a constructive well-founded relation. An element is said to be accessible if all strictly smaller elements are accessible. A relation is then well-founded if all elements are accessible. This is formulated as follows:

```

data Acc {A : Set a}
  ( _<_ : A → A → Set ℓ )
  ( x : A )
  : Set (a ⊔ ℓ) where
  acc : (rs : ∀ y → y < x → Acc _<_ y) → Acc _<_ x

```

```

WellFounded : (A → A → Set ℓ) → Set _
WellFounded _<_ = ∀ x → Acc _<_ x

```

That is, an element of a type is accessible for a relation if all strictly smaller elements of it are also accessible. A relation is well-founded if all values are accessible with respect to that relation. This can then be used to define induction with arbitrary recursive calls on smaller values:

```

wfRec : (P : A → Set ℓ)
  → (∀ x → (∀ y → y < x → P y) → P x)
  → ∀ x → P x

```

The `wfRec` function is defined using structural recursion on an argument of type `Acc`, so the type checker accepts it. Well-founded induction computes a fixed point of the function, meaning that the particular proof that the strict order holds is irrelevant:

```

unfold-wfRec : ∀ {x}
  → wfRec P f x ≡ f x (λ y _ → wfRec P f y)

```

Following the construction of [Kraus et al. \[2023\]](#), we can show that the strict ordering on Brouwer trees is well-founded. First, we prove a lemma: if a value is accessible, then all (not necessarily strictly) smaller terms are also accessible.

```

smaller-accessible : (x : Tree)
  → Acc _<_ x → ∀ y → y ≤ x → Acc _<_ y
smaller-accessible x (acc r) y x≤y
  = acc (λ y' y'<y → r y' (<-≤-in-< y'<y x≤y))

```


Then structural induction shows that all terms are accessible. The key observations are that zero is trivially accessible, since no trees are strictly smaller than it, and that the only way to derive $\uparrow t \leq (\text{Lim } c f)$ is with $\leq\text{-cocone}$, yielding a concrete k for which $\uparrow t \leq f k$, on which we recur.

```
ordWF : WellFounded _<_
ordWF Z = acc λ _ ()
ordWF (↑ x)
  = acc (λ { y (≤-sucMono y ≤ x)
    → smaller-accessible x (ordWF x) y y ≤ x })
ordWF (Lim c f) = acc wfLim
where
  wfLim : (y : Tree) → (y < Lim c f) → Acc _<_ y
  wfLim y (≤-cocone f k y < f k)
    = smaller-accessible (f k)
      (ordWF (f k)) y (<-in-≤ y < f k)
```

This lets us use Brouwer trees as the decreasing metric for well-founded recursion. However, the `wfRec` function only worked with one argument. To handle recursion with more than one argument, we need a way to combine ordinals.

3 First Attempts at a Join

One way to do induction over multiple arguments is to do well-founded induction over the maximum of the sizes of those arguments. Doing this requires a maximum function, or in semilattice terminology, a join operator.

Here we present two faulty implementations of a join operator for Brouwer trees. The first uses limits to define the join, but does not satisfy strict monotonicity. The second is defined inductively. It satisfies strict monotonicity, but fails to be the least of all upper bounds, and requires us to assume that limits are only taken over non-empty types. In Section 4, we define SMB-trees: a refinement of Brouwer trees with the benefits of both versions of the maximum.

3.1 Limit-based Maximum

Since the limit constructor finds the least upper bound of the image of a function, it should be possible to define the maximum of two trees as a special case of general limits. Indeed, we can compute the maximum of t_1 and t_2 as the limit of the function that produces t_1 when given 0 and t_2 otherwise.

```
limMax : Tree → Tree → Tree
limMax t1 t2 = NLim λ n → if0 n t1 t2
```

This version of the maximum has the properties we want from a maximum function: it is an upper bound on its arguments, and it is idempotent.

```
limMax≤L : ∀ {t1 t2} → t1 ≤ limMax t1 t2
limMax≤L {t1} {t2}
```

```
= ≤-cocone _ (Iso.inv CNIso 0)
(subst
  (λ x → t1 ≤ if0 x t1 t2)
  (sym (Iso.rightInv CNIso 0))
  (≤-refl t1))
```

```
limMax≤R : ∀ {t1 t2} → t2 ≤ limMax t1 t2
```

```
limMaxIdem : ∀ {t} → limMax t t ≤ t
```

```
limMaxIdem {t} = ≤-limiting _ helper
```

where

```
helper : ∀ k → if0 (Iso.fun CNIso k) t t ≤ t
helper k with Iso.fun CNIso k
... | zero = ≤-refl t
... | suc n = ≤-refl t
```

From these properties, we can compute several other useful properties: monotonicity, commutativity, and that it is in fact the least of all upper bounds.

```
limMaxMono : ∀ {t1 t2 t'1 t'2}
```

```
→ t1 ≤ t'1 → t2 ≤ t'2
```

```
→ limMax t1 t2 ≤ limMax t'1 t'2
```

```
limMaxCommut : ∀ {t1 t2} → limMax t1 t2 ≤ limMax t2 t1
```

```
limMaxLUB : ∀ {t1 t2 t} → t1 ≤ t → t2 ≤ t → limMax t1 t2 ≤ t
```

This version of the maximum is a least upper bound: by definition `Lim` denotes the least upper bound of a function's image, and `limMax` is simply `Lim` applied to a function whose image has (at most) two elements.

Limitation: Strict Monotonicity The one crucial property that this formulation lacks is that it is not strictly monotone: we cannot deduce $\text{max } t_1 t_2 < \text{max } t'_1 t'_2$ from $t_1 < t'_1$ and $t_2 < t'_2$. This is because the only way to construct a proof that $\uparrow t \leq \text{Lim } c f$ is using the $\leq\text{-cocone}$ constructor. So we would need to prove that $\uparrow(\text{max } t_1 t_2) \leq t'_1$ or that $\uparrow(\text{max } t_1 t_2) \leq t'_2$, which cannot be deduced from the premises alone. What we want is to have $\uparrow \text{max } t_1 t_2 \leq \text{max}(\uparrow t_1)(\uparrow t_2)$, so that strict monotonicity is a direct consequence of ordinary monotonicity of the maximum. This is not possible when defining the constructor as a limit.

3.2 Recursive Maximum

In our next attempt at defining a maximum operator, we obtain strict monotonicity by making `indMax` ($\uparrow t_1$) ($\uparrow t_2$) = $\uparrow(\text{indMax } t_1 t_2)$ hold definitionally. Then, provided `indMax` is monotone, it will also be strictly monotone.

To do this, we compute the maximum of two trees recursively, pattern matching on the operands. We use a *view* [McBride and McKinna 2004] to identify the cases we are matching on: we are matching on two arguments, each with three possible constructors, but several cases overlap. Using a view type lets us avoid enumerating all nine possibilities when defining the maximum and proving its properties.

To begin, we parameterize our definition over a function yielding some element for any code's type. Having a representative of every code will be useful in computing the maximum of a limit and some other tree, since we do not need to handle the special case where the limit of an empty sequence is zero.

```
module IndMax {ℓ}
  (C : Set ℓ)
  (El : C → Set ℓ)
  (CN : C) (CNIso : Iso (El CN) ℕ)
  (default : (c : C) → El c) where
```

We then define our view type:

```
private
  data IndMaxView : Tree → Tree → Set ℓ where
    IndMaxZ-L : ∀ {t} → IndMaxView Z t
    IndMaxZ-R : ∀ {t} → IndMaxView t Z
    IndMaxLim-L : ∀ {t} {c : C} {f : El c → Tree}
      → IndMaxView (Lim c f) t
    IndMaxLim-R : ∀ {t} {c : C} {f : El c → Tree}
      → (∀ {c' : C} {f' : El c' → Tree} → ¬ (t = Lim c' f'))
      → IndMaxView t (Lim c f)
    IndMaxLim-Suc : ∀ {t1 t2} → IndMaxView (↑ t1) (↑ t2)
```

opaque

```
indMaxView : ∀ t1 t2 → IndMaxView t1 t2
```

Our view type has five cases. The first two cases handle when either input is zero, and the next two cases handle when either input is a limit. The final case is when both inputs are successors. The helper `indMaxView` computes the view for any pair of trees.

The maximum is then defined by pattern matching on the view for its arguments:

```
indMax : Tree → Tree → Tree
indMax' : ∀ {t1 t2} → IndMaxView t1 t2 → Tree
```

```
indMax t1 t2 = indMax' (indMaxView t1 t2)
indMax' {Z} {t2} IndMaxZ-L = t2
indMax' {t1} {Z} IndMaxZ-R = t1
indMax' {Lim c f} {t2} IndMaxLim-L
  = Lim c λ x → indMax (f x) t2
indMax' {t1} {Lim c f} (IndMaxLim-R _)
  = Lim c (λ x → indMax t1 (f x))
indMax' {↑ t1} {↑ t2} IndMaxLim-Suc = ↑ (indMax t1 t2)
```

The maximum of zero and t is always t , and the maximum of t and the limit of f is the limit of the function computing the maximum between t and $f x$. Finally, the maximum of two successors is the successor of the two maxima, giving the definitional equality we need for strict monotonicity.

This definition only works when limits of all codes are inhabited. The `≤-limiting` constructor means that $\text{Lim } c f \leq Z$ whenever $\text{El } c$ is uninhabited. So `indMax ↑Z Lim c f` will

not actually be an upper bound for $\uparrow Z$ if c has no inhabitants. In Section 4.2 we show how to circumvent this restriction.

Under the assumption that all code are inhabited, we obtain several of our desired properties for a maximum: it is an upper bound, it is monotone and strictly monotone, and it is associative and commutative. The proof bodies are omitted: they are straightforward reasoning by cases, but they are long and tedious.

opaque

```
unfolding indMax indMax'
```

```
indMax-≤L : ∀ {t1 t2} → t1 ≤ indMax t1 t2
indMax-≤L {t1} {t2} with indMaxView t1 t2
... | IndMaxZ-L = ≤-Z
... | IndMaxZ-R = ≤-refl _
... | IndMaxLim-L {f = f'}
  = extLim f (λ x → indMax (f x) t2) (λ k → indMax-≤L)
... | IndMaxLim-R {f = f'} _
  = underLim λ k → indMax-≤L {t2 = f k}
... | IndMaxLim-Suc
  = ≤-sucMono indMax-≤L
```

```
indMax-≤R : ∀ {t1 t2} → t2 ≤ indMax t1 t2
```

```
indMax-monoL : ∀ {t1 t'1 t2}
  → t1 ≤ t'1 → indMax t1 t2 ≤ indMax t'1 t2
indMax-monoR : ∀ {t1 t2 t'2}
  → t2 ≤ t'2 → indMax t1 t2 ≤ indMax t1 t'2
```

```
indMax-mono : ∀ {t1 t2 t'1 t'2}
  → t1 ≤ t'1 → t2 ≤ t'2 → indMax t1 t2 ≤ indMax t'1 t'2
```

```
indMax-strictMono : ∀ {t1 t2 t'1 t'2}
  → t1 < t'1 → t2 < t'2 → indMax t1 t2 < indMax t'1 t'2
```

```
indMax-strictMono lt1 lt2 = indMax-mono lt1 lt2
```

```
indMax-assocL : ∀ t1 t2 t3
  → indMax t1 (indMax t2 t3) ≤ indMax (indMax t1 t2) t3
```

```
indMax-assocR : ∀ t1 t2 t3
  → indMax (indMax t1 t2) t3 ≤ indMax t1 (indMax t2 t3)
```

```
indMax-commut : ∀ t1 t2
  → indMax t1 t2 ≤ indMax t2 t1
```

Limitation: Idempotence The problem with an inductive definition of the maximum is that we cannot prove that it is idempotent. Since `indMax` is associative and commutative, proving idempotence is equivalent to proving that it computes a true least-upper-bound.

The difficulty lies in showing that `indMax (Lim c f) (Lim c f) ≤ (Lim c f)`. By our definition, `indMax (Lim c f) (Lim c f)` reduces to:

```
(Lim c λ x → (Lim c (λ y → indMax (f x) (f y)))) ≤ Lim c f
```

We cannot use $\leq\text{-cocone}$ to prove this, since the left-hand side is not necessarily equal to $f\ k$ for any $k : El\ c$. So the only possibility is to use $\leq\text{-limiting}$. Applying it twice, along with a use of commutativity of indMax , we are left with the following goal:

$$\forall x \rightarrow \forall y \rightarrow (\text{indMax}\ (f\ x)\ (f\ y) \leq \text{Lim}\ c\ f)$$

There is no a priori way to prove this goal without already having a proof that indMax is a least upper bound. But proving that was the whole point of proving idempotence! An inductive hypothesis would give that $\text{indMax}\ (f\ x)\ (f\ x) \leq f\ x \leq \text{Lim}\ c\ f$, but it does not apply when the arguments to indMax are not equal. Because we are working with constructive ordinals, we have no trichotomy property [Kraus et al. 2023], and hence no guarantee that $\text{indMax}\ (f\ x)\ (f\ y)$ will be one of $f\ x$ and $f\ y$.

We now have two competing definitions for the maximum: the limit version, which is not strictly monotone, and the inductive version, which is not actually a least upper bound. In the next section, we describe a large class of trees for which indMax is idempotent, and hence does compute a true upper bound. We then use that in Section 4.2 to create a version of ordinals whose join has the best properties of both limMax and indMax .

4 Trees with a Strictly-Monotone Idempotent Join

4.1 Well-Behaved Trees

Our first step in defining an ordinal notation with a well-behaved maximum is to identify a class of Brouwer trees which are well-behaved with respect to the inductive maximum. As we saw in the previous section, neither the limit based nor the inductive definition of the maximum was satisfactory.

The solution, it turns out, is more limits: if we indMax a term with itself an infinite number of times, the result will be idempotent with respect to indMax . This is essentially an application of Kleene's Fixed-Point Theorem [Cousot and Cousot 1979], using transfinite iteration to find the solution to $x \approx \text{indMax}\ x\ x$.

First, we define a function to indMax a term with itself n times or a given number n :

$$\begin{aligned} \text{nindMax} &: \text{Tree} \rightarrow \mathbb{N} \rightarrow \text{Tree} \\ \text{nindMax}\ t\ \mathbb{N}.\text{zero} &= Z \\ \text{nindMax}\ t\ (\mathbb{N}.\text{suc}\ n) &= \text{indMax}\ (\text{nindMax}\ t\ n)\ t \end{aligned}$$

To compute a tree equivalent to the infinite chain of applications $\text{indMax}\ t\ (\text{indMax}\ t\ (\text{indMax}\ t\ \dots))$, we take the limit of n applications over all n :

$$\begin{aligned} \text{indMax}\infty &: \text{Tree} \rightarrow \text{Tree} \\ \text{indMax}\infty\ t &= \mathbb{N}\text{Lim}\ (\lambda n \rightarrow \text{nindMax}\ t\ n) \end{aligned}$$

This operator has useful basic properties: it is monotone, and it computes an upper bound on its argument.

$$\text{indMax}\infty\text{-self} : \forall t \rightarrow t \leq \text{indMax}\infty\ t$$

$$\begin{aligned} \text{indMax}\infty\text{-mono} &: \forall \{t_1\ t_2\} \\ &\rightarrow t_1 \leq t_2 \\ &\rightarrow (\text{indMax}\infty\ t_1) \leq (\text{indMax}\infty\ t_2) \end{aligned}$$

However, the most important property that we want from $\text{indMax}\infty$ is that indMax is idempotent with respect to it. The first step to showing this is realizing that we can take the maximum of t and $\text{indMax}\infty\ t$, and that we have a tree that is no larger than $\text{indMax}\infty\ t$: because it is already an infinite chain of applications, adding one more makes no difference.

$$\begin{aligned} \text{indMax}\infty\text{-ltn1} &: \forall t \rightarrow \text{indMax}\ (\text{indMax}\infty\ t)\ t \leq \text{indMax}\infty\ t \\ \text{indMax}\infty\text{-ltn1}\ t &= \leq\text{-limiting}\ _\lambda k \rightarrow \text{helper}\ (\text{Iso.fun}\ \text{CNIso}\ k) \\ \text{where} \\ \text{helper} &: \forall n \rightarrow \text{indMax}\ (\text{nindMax}\ t\ n)\ t \leq \text{indMax}\infty\ t \\ \text{helper}\ n &= \\ &\leq\text{-cocone}\ _\lambda (\text{Iso.inv}\ \text{CNIso}\ (\mathbb{N}.\text{suc}\ n)) \\ &(\text{subst}\ (\lambda sn \rightarrow \text{nindMax}\ t\ (\mathbb{N}.\text{suc}\ n) \leq \text{nindMax}\ t\ sn) \\ &(\text{sym}\ (\text{Iso.rightInv}\ \text{CNIso}\ (\text{suc}\ n))) \\ &(\leq\text{-refl}\ _)) \end{aligned}$$

If adding one more $\text{indMax}\ t$ has no effect, then adding n more will also have no effect:

$$\begin{aligned} \text{indMax}\infty\text{-ltn} &: \forall n\ t \\ &\rightarrow \text{indMax}\ (\text{indMax}\infty\ t)\ (\text{nindMax}\ t\ n) \leq \text{indMax}\infty\ t \\ \text{indMax}\infty\text{-ltn}\ \mathbb{N}.\text{zero}\ t &= \text{indMax}\leq Z\ (\text{indMax}\infty\ t) \\ \text{indMax}\infty\text{-ltn}\ (\mathbb{N}.\text{suc}\ n)\ t &= \\ \text{indMax}\text{-monoR} \\ &\{t_1 = \text{indMax}\infty\ t\} (\text{indMax}\text{-commut}\ (\text{nindMax}\ t\ n)\ t) \\ &\leq \S\ \text{indMax}\text{-assocL}\ (\text{indMax}\infty\ t)\ t\ (\text{nindMax}\ t\ n) \\ &\leq \S\ \text{indMax}\text{-monoL}\ (\text{indMax}\infty\text{-ltn1}\ t) \\ &\leq \S\ \text{indMax}\infty\text{-ltn}\ n\ t \end{aligned}$$

It remains to show that taking indMax of $\text{indMax}\infty\ t$ with itself does not make it larger. By our inductive definition of indMax , we have that

$$\text{indMax}\ (\text{indMax}\infty\ t)(\text{indMax}\infty\ t)$$

is equal to

$$\mathbb{N}\text{Lim}\ (\lambda n \rightarrow \text{indMax}\ (\text{nindMax}\ n\ t)\ (\text{indMax}\infty\ t))$$

Our previous lemma gives that, for any n , $\text{indMax}\infty\ t$ is an upper bound for $\text{indMax}\ (\text{nindMax}\ n\ t)\ (\text{indMax}\infty\ t)$. So $\leq\text{-limiting}$ gives that the limit over all n is also bounded by $\text{indMax}\infty\ t$, i.e. Lim constructs the least of all upper bounds. This gives us our key result: up to \leq , indMax is idempotent on values constructed with $\text{indMax}\infty$.

$$\begin{aligned} \text{indMax}\infty\text{-idem} &: \forall t \\ &\rightarrow \text{indMax}\ (\text{indMax}\infty\ t)\ (\text{indMax}\infty\ t) \leq \text{indMax}\infty\ t \\ \text{indMax}\infty\text{-idem}\ t &= \\ &\leq\text{-limiting}\ _\lambda k \rightarrow \\ &(\text{indMax}\text{-commut} \end{aligned}$$

```
(nindMax t (Iso.fun CNIso k)) (indMax∞ t))
≤ § indMax-∞ltN (Iso.fun CNIso k) t
```

There is one last property to prove that will be useful in the next section: `indMax∞ t` is a lower bound on t , and hence equivalent to it, whenever `indMax` is idempotent on t . If taking `indMax` of t with itself does not increase its size, doing so n times will not increase its size, so again the result follows from `Lim` being the least upper bound.

```
indMax∞-≤ : ∀ {t} → indMax t t ≤ t → indMax∞ t ≤ t
indMax∞-≤ lt
= ≤-limiting _
  λ k → nindMax-≤ (Iso.fun CNIso k) lt
where
  nindMax-≤ : ∀ {t} n → indMax t t ≤ t → nindMax t n ≤ t
  nindMax-≤ N.zero lt = ≤-Z
  nindMax-≤ {t = t} (N.suc n) lt
    = indMax-monoL (nindMax-≤ n lt)
    ≤ § lt
```

An immediate corollary of this is that `indMax∞ (indMax∞ t)` is equivalent to `indMax∞ t`.

4.2 Strictly Monotone Brouwer Trees

Now that we have identified a substantial class of well-behaved Brouwer trees, we want to define a new type containing only those trees. In this section, we will define strictly monotone Brouwer trees (SMB-trees), and show how they can be given a similar interface to Brouwer trees.

To begin, we declare a new Agda module, with the same parameters we have been working with thus far: a type of codes, interpretations of those codes into types, and a code whose interpretation is isomorphic to \mathbb{N} .

```
module SMBTree {ℓ}
  (C : Set ℓ)
  (El : C → Set ℓ)
  (CN : C) (CNIso : Iso (El CN) N) where
```

Next we import all of our definitions so far, using the “Brouwer” prefix to distinguish them from the trees and ordering we are about to define. Critically, we do not instantiate these with the same interpretation function. Instead, we interpret each code wrapped in `Maybe`. Note that if a type T is isomorphic to \mathbb{N} , then `Maybe T` is as well. Wrapping in `Maybe` ensures that we always take Brouwer limits over non-empty sets, an assumption that was critical for the definitions of Section 3.2. Essentially, we are adding an explicit zero to every sequence whose limit we take, so that the sequences are never empty, but the upper bound does not change. This detail is hidden in the interface for SMB-trees: the assumption of non-emptiness is only used in the Brouwer trees underlying SMB-trees.

```
import Brouwer
C
(λ c → Maybe (El c))
CN (maybeNatIso CNIso)
as Brouwer
```

Refining Brouwer Trees We define SMB-trees as a dependent record, containing an underlying Brouwer tree, and a proof that `indMax` is idempotent on this tree.

```
record SMBTree : Set ℓ where
  constructor MkTree
  field
    rawTree : Brouwer.Tree
    isIdem : (indMax rawTree rawTree) Brouwer.≤ rawTree
open SMBTree
```

We can then define so-called “smart-constructors” corresponding to each of the constructors for Brouwer-trees: zero, successor, and limit. Zero and successor directly correspond to the Brouwer tree zero and successor. Their proofs of idempotence are trivial from the properties of Brouwer `≤`.

```
opaque
  unfolding indMax

  Z : SMBTree
  Z = MkTree Brouwer.Z Brouwer.≤-Z

  ↑ : SMBTree → SMBTree
  ↑ (MkTree t pf)
    = MkTree (Brouwer.↑ t) (Brouwer.≤-sucMono pf)
```

However, constructing the limit of a sequence of SMB-trees is not so easy. Since we instantiated `El` to wrap its result in `Maybe`, we need to handle `nothing` for each limit, but we can use `Z` as a default value, since adding it to any sequence does not change the least upper bound. More challenging is how, as we saw in Section 3.2, Brouwer trees do not have `indMax (Lim c f) (Lim c f) ≤ Lim c f`, so we cannot directly produce a proof of idempotence.

Our key insight is to define limits of SMB-trees using `indMax∞` on the underlying trees: for any function producing SMB-trees, we take the limit of the underlying trees, then `indMax` that result with itself an infinite number of times. The idempotence proof is then the property of `indMax∞` that we proved in Section 4.1.

```
Lim : ∀ (c : C) → (f : El c → SMBTree) → SMBTree
Lim c f =
  MkTree
    (indMax∞
      (Brouwer.Lim c
        (maybe' (λ x → rawTree (f x)) Brouwer.Z)))
    (indMax∞-idem _)
```


Ordering SMB-trees SMB-trees are ordered by the order on their underlying Brouwer trees:

```
record _≤_ (t1 t2 : SMBTree) : Set ℓ where
  constructor mk≤
  inductive
  field
  get≤ : (rawTree t1) Brouwer.≤ (rawTree t2)
open _≤_
```

The successor function allows us to define a strict order on SMB-trees.

```
_<_ : SMBTree → SMBTree → Set ℓ
_<_ t1 t2 = (↑ t1) ≤ t2
```

The next step is to prove that our SMB-tree constructors satisfy the same inequalities as Brouwer trees. Since SMB-trees are ordered by their underlying Brouwer trees, most properties can be directly lifted from Brouwer trees to SMB-trees.

```
opaque
  unfolding Z ↑
  ≤↑ : ∀ t → t ≤ ↑ t
  ≤↑ t = mk≤ (Brouwer.≤↑ t _)

  _≤_ : ∀ {t1 t2 t3} → t1 ≤ t2 → t2 ≤ t3 → t1 ≤ t3
  _≤_ (mk≤ lt1) (mk≤ lt2) = mk≤ (Brouwer.≤-trans lt1 lt2)

  _≤-refl : ∀ {t} → t ≤ t
  _≤-refl = mk≤ (Brouwer.≤-refl _)
```

The constructors for \leq each have a counterpart for SMB-trees. For zero and successor, these are trivially lifted.

```
≤-Z : ∀ {t} → Z ≤ t
≤-Z = mk≤ Brouwer.≤-Z

≤-sucMono : ∀ {t1 t2} → t1 ≤ t2 → ↑ t1 ≤ ↑ t2
≤-sucMono (mk≤ lt) = mk≤ (Brouwer.≤-sucMono lt)
```

The constructors for ordering limits require more attention. To show that an SMB-tree limit is an upper bound, we use the fact that the underlying limit was an upper bound, and the fact that $\text{indMax}\infty$ is as large as its argument, since the SMB-tree Lim wraps its result in $\text{indMax}\infty$. Note that, since we already have transitivity for our new \leq , we can simply show that $f k$ is less than the limit of f , avoiding the more complicated form of $\leq\text{-cocone}$.

```
≤-limUpperBound : ∀ {c : C} → {f : El c → SMBTree}
  → ∀ k → f k ≤ Lim c f
≤-limUpperBound {c = c} {f = f} k
  = mk≤ (Brouwer.≤-cocone _ (just k) (Brouwer.≤-refl _))
    Brouwer.≤ ; indMax∞-self (Brouwer.Lim c _)
```

Finally, we need to show that the SMB-tree limit is less than all other upper bounds. Suppose $t : \text{SMBTree}$ is an upper bound for f , and t_u is the underlying tree for t , and f_u computes the underlying trees for f . Then $\leq\text{-limiting}$ gives that the underlying tree for t is an upper bound for the trees underlying the image of f . However, the SMB-tree limit wraps its result in $\text{indMax}\infty$, so we need to show that $\text{indMax}\infty$ of the limit is also less than t' . The monotonicity of $\text{indMax}\infty$ then gives that $\text{indMax}(\text{Lim } c f_u)$ is less than $\text{indMax}\infty t'$. In Section 4.1, we showed that $\text{indMax}\infty$ had no effect on Brouwer trees that indMax was idempotent on. This is exactly what the isIdem field of SMB-trees contains! So we have $\text{indMax}\infty t' \leq t'$, and transitivity gives our result.

```
≤-limLeast : ∀ {c : C} → {f : El c → SMBTree}
  → {t : SMBTree}
  → (∀ k → f k ≤ t) → Lim c f ≤ t
≤-limLeast {f = f} {t = MkTree t idem} lt
  = mk≤ (
    indMax∞-mono
    (Brouwer.≤-limiting _
     (maybe (λ k → get≤ (lt k)) Brouwer.≤-Z))
    Brouwer.≤ ; (indMax∞-≤ idem) )
```

The Join for SMB-trees Our whole reason for defining SMB-trees was to define a well-behaved maximum operator, and we finally have the tools to do so. We can define the join in terms of indMax on the underlying trees. The proof that the indMax is idempotent on the result follows from associativity, commutativity, and monotonicity of indMax .

```
opaque
  unfolding indMax Z ↑ indMaxView
  max : SMBTree → SMBTree → SMBTree
  max t1 t2 =
    MkTree
    (indMax (rawTree t1) (rawTree t2))
    (indMax-swap4 {t1 = rawTree t1} {t1' = rawTree t2}
      {t2 = rawTree t1} {t2' = rawTree t2}
      Brouwer.≤ ; indMax-mono (isIdem t1) (isIdem t2))
```

For Brouwer trees, indMax had all the properties we wanted except for idempotence. All of these can be lifted directly to SMB-trees:

```
max-≤L : ∀ {t1 t2} → t1 ≤ max t1 t2
max-≤R : ∀ {t1 t2} → t2 ≤ max t1 t2
max-mono : ∀ {t1 t1' t2 t2'} → t1 ≤ t1' → t2 ≤ t2' →
  max t1 t2 ≤ max t1' t2'
max-idem≤ : ∀ {t} → t ≤ max t t
max-commut : ∀ t1 t2 → max t1 t2 ≤ max t2 t1
```

$\text{max-assocL} : \forall t_1 t_2 t_3$
 $\rightarrow \text{max } t_1 (\text{max } t_2 t_3) \leq \text{max } (\text{max } t_1 t_2) t_3$

$\text{max-assocR} : \forall t_1 t_2 t_3$
 $\rightarrow \text{max } (\text{max } t_1 t_2) t_3 \leq \text{max } t_1 (\text{max } t_2 t_3)$

In particular, max is strictly monotone, and distributes over the successor:

$\text{max-strictMono} : \forall \{t_1 t'_1 t_2 t'_2 : \text{SMBTree}\}$
 $\rightarrow t_1 < t'_1 \rightarrow t_2 < t'_2 \rightarrow \text{max } t_1 t_2 < \text{max } t'_1 t'_2$

$\text{max-sucMono} : \forall \{t_1 t_2 t'_1 t'_2 : \text{SMBTree}\}$
 $\rightarrow \text{max } t_1 t_2 \leq \text{max } t'_1 t'_2 \rightarrow \text{max } t_1 t_2 < \text{max } (\uparrow t'_1) (\uparrow t'_2)$

However, because we restricted SMB-trees to only contain Brouwer trees that indMax is idempotent on, we can prove that Max is idempotent for SMB-trees:

$\text{max-idem} : \forall \{t : \text{SMBTree}\} \rightarrow \text{max } t t \leq t$
 $\text{max-idem } \{t = \text{MkTree } t \text{ pf}\} = \text{mk} \leq \text{pf}$

These together are enough to prove that our maximum is the least of all upper bounds.

$\text{max-LUB} : \forall \{t_1 t_2 t\} \rightarrow t_1 \leq t \rightarrow t_2 \leq t \rightarrow \text{max } t_1 t_2 \leq t$
 $\text{max-LUB } lt1 lt2 = \text{max-mono } lt1 lt2 \leq \text{max-idem}$

Perhaps surprisingly, this means that an SMB-tree version of limMax is equivalent to max , since they are both the least upper bound. This in turn means that the limit based maximum is strictly monotone for SMB-trees.

$\text{NLim} : (\mathbb{N} \rightarrow \text{SMBTree}) \rightarrow \text{SMBTree}$
 $\text{NLim } f = \text{Lim } \text{CN } (\lambda \text{ cn} \rightarrow f (\text{Iso.fun } \text{CNIso } \text{cn}))$

$\text{max}' : \text{SMBTree} \rightarrow \text{SMBTree} \rightarrow \text{SMBTree}$
 $\text{max}' t_1 t_2 = \text{NLim } (\lambda n \rightarrow \text{if0 } n t_1 t_2)$

$\text{max}'\text{-}\leq\text{L} : \forall \{t_1 t_2\} \rightarrow t_1 \leq \text{max}' t_1 t_2$

$\text{max}'\text{-}\leq\text{R} : \forall \{t_1 t_2\} \rightarrow t_2 \leq \text{max}' t_1 t_2$

$\text{max}'\text{-LUB} : \forall \{t_1 t_2 t\} \rightarrow t_1 \leq t \rightarrow t_2 \leq t \rightarrow \text{max}' t_1 t_2 \leq t$

$\text{max}\leq\text{max}' : \forall \{t_1 t_2\} \rightarrow \text{max } t_1 t_2 \leq \text{max}' t_1 t_2$
 $\text{max}\leq\text{max}' = \text{max-LUB } \text{max}'\text{-}\leq\text{L } \text{max}'\text{-}\leq\text{R}$

$\text{max}'\leq\text{max} : \forall \{t_1 t_2\} \rightarrow \text{max}' t_1 t_2 \leq \text{max } t_1 t_2$
 $\text{max}'\leq\text{max} = \text{max}'\text{-LUB } \text{max}'\text{-}\leq\text{L } \text{max}'\text{-}\leq\text{R}$

Well-founded Ordering on SMB-trees Our motivation for defining SMB-trees was defining well-founded recursion, so the final piece of our definition is a proof that the strict ordering of SMB-trees is well-founded. Intuitively this should hold: there are no infinite descending chains of Brouwer trees, and there are fewer SMB-trees than Brouwer trees, so there can be no infinite descending chains of SMB-trees. The key lemma is that an SMB-tree is accessible if its underlying Brouwer tree is.

$\text{sizeWF} : \text{WellFounded } _< _$
 $\text{sizeWF } t = \text{sizeAcc } (\text{Brouwer.ordWF } (\text{rawTree } t))$
 where
 $\text{sizeAcc} : \forall \{t\}$
 $\rightarrow \text{Acc Brouwer.}_< _ (\text{rawTree } t)$
 $\rightarrow \text{Acc } _< _ t$
 $\text{sizeAcc } \{t\} (\text{acc } x)$
 $= \text{acc } ((\lambda y \text{ lt} \rightarrow \text{sizeAcc } (x (\text{rawTree } y) (\text{get} \leq \text{lt}))))$

Thus, we have an ordinal type with limits, a strictly monotone join, and well-founded recursion.

5 An Algebraic Perspective

As a final contribution, we give an algebraic viewpoint of SMBTrees, in terms of equivalences rather than orderings. There are no new results in this section, but the equational view highlights the ways in which a strictly monotone join is useful in reasoning.

SMB-trees cannot be completely characterized using first order equations, since Lim is an infinitary operator. Nevertheless, we anticipate that many of the equations here could be useful in developing automated rewriting tools or tactics for reasoning about SMBTrees. The constraint of $t_1 < t_2$ can be translated into the equation $\uparrow t_1 \vee t_2 \approx t_2$, which can then be mechanically simplified according to the equations in the following sections.

5.1 Semilattices and Setoids

Unfortunately, SMB-trees are only a preorder, not a partial order. Because we are working in vanilla Agda, we have no function extensionality, so applying Lim to definitionally distinct but extensionally equal functions produces trees that are equivalent but not equal. Postulating that equivalent terms are equal would be inconsistent: inductive and limit-based joins are equivalent for SMB-trees, so $\uparrow (t_1 \vee t_2)$ is equivalent to $\text{limMax } (\uparrow t_1) (\uparrow t_2)$, even though their heads are distinct datatype constructors. Likewise, $\text{Lim } c f$ is equivalent to Z any time $\text{El } c$ is uninhabited.

As such, we present our equations in the setoid style i.e. up to an equivalence relation, but the results in this section could be adapted to quotient types in a system like Cubical Agda [Vezzosi et al. 2019]. First, we establish that SMB-trees are a bounded join-semilattice.

$\text{TreeSemiLat} : \text{BoundedJoinSemilattice } \ell \ell \ell$

$\text{Carrier TreeSemiLat} = \text{SMBTree}$

$_ \approx _ \text{TreeSemiLat } t_1 t_2 = t_1 \text{ SMBTree.}\leq t_2 \times t_2 \text{ SMBTree.}\leq t_1$

$_ \leq _ \text{TreeSemiLat} = \text{SMBTree.}_ \leq _$

$_ \vee _ \text{TreeSemiLat} = \text{SMBTree.max}$

$\perp \text{TreeSemiLat} = \text{SMBTree.Z}$

— ...

Orderings between trees can then be expressed equationally using the join: t_1 is smaller than t_2 iff their join is t_2 .

$\text{ord} \rightarrow \text{equiv} : \forall \{t_1 \ t_2\} \rightarrow t_1 \leq t_2 \rightarrow t_1 \vee t_2 \approx t_2$
 $\text{equiv} \rightarrow \text{ord} : \forall \{t_1 \ t_2\} \rightarrow t_1 \vee t_2 \approx t_2 \rightarrow t_1 \leq t_2$

This means that our ordering respects equivalence. Additionally, the successor, join and limit constructors are congruences for our equivalence: equivalent inputs produce equivalent outputs. These can be combined with the proof irrelevance of well-founded recursion to rewrite ordering goals according to algebraic laws.

$\leq \approx : \forall \{t_1 \ t_2 \ s_1 \ s_2\}$
 $\rightarrow s_1 \leq t_1 \rightarrow s_1 \approx s_2 \rightarrow t_1 \approx t_2 \rightarrow s_2 \leq t_2$
 $< \approx : \forall \{t_1 \ t_2 \ s_1 \ s_2\}$
 $\rightarrow s_1 < t_1 \rightarrow s_1 \approx s_2 \rightarrow t_1 \approx t_2 \rightarrow s_2 < t_2$
 $\uparrow\text{-cong} : \forall \{t_1 \ t_2\}$
 $\rightarrow t_1 \approx t_2 \rightarrow \uparrow t_1 \approx \uparrow t_2$
 $\text{Lim-cong} : \forall \{c\} \{f_1 \ f_2\}$
 $\rightarrow (\forall x \rightarrow f_1 \ x \approx f_2 \ x) \rightarrow \text{Lim } c \ f_1 \approx \text{Lim } c \ f_2$
 $\text{max-cong} : \forall \{s_1 \ s_2 \ t_1 \ t_2\}$
 $\rightarrow s_1 \approx s_2 \rightarrow t_1 \approx t_2 \rightarrow s_1 \vee t_1 \approx s_2 \vee t_2$

This gives us a framework to present the properties of SMB-trees equationally. For instance, the semilattice properties of the join can be given algebraically: a semilattice is a commutative, idempotent semigroup.

$\text{assoc} : \forall \{t_1 \ t_2 \ t_3\} \rightarrow t_1 \vee (t_2 \vee t_3) \approx (t_1 \vee t_2) \vee t_3$
 $\text{commut} : \forall \{t_1 \ t_2\} \rightarrow t_1 \vee t_2 \approx t_2 \vee t_1$
 $\text{idem} : \forall \{t\} \rightarrow t \vee t \approx t$

5.2 Successor: The Inflationary Endomorphism

The algebraic version of strict monotonicity is that the successor function is what [Bezem and Coquand \[2022\]](#) call an *inflationary endomorphism*, i.e. a unary operator whose interactions with the join behave like the successor on natural numbers. To our knowledge, SMB-trees are the first ordinal notation in type theory for which the successor is inflationary and arbitrary limits are supported.

There are two laws to inflationary endomorphisms. First, the maximum of $\uparrow t$ and t must be $\uparrow t$, which captures the idea that t is less than $\uparrow t$.

$\uparrow\text{absorb} : \forall \{t\} \rightarrow t \vee (\uparrow t) \approx \uparrow t$
 $\uparrow\text{absorb} =$
 $\text{max-mono } (\leq \uparrow _) \leq\text{-refl } \leq ; \text{max-idem}$
 $, \text{max-}\leq\text{R}$

Second, the successor must distribute over the join. Recall that this was precisely the condition we used to establish strict monotonicity.

$\uparrow\text{dist} : \forall \{t_1 \ t_2\} \rightarrow \uparrow (t_1 \vee t_2) \approx \uparrow t_1 \vee \uparrow t_2$
 $\uparrow\text{dist } \{t_1\} \{t_2\} =$
 $\text{max-sucMono } \leq\text{-refl}$
 $, \text{max-LUB } (\leq\text{-sucMono max-}\leq\text{L}) (\leq\text{-sucMono max-}\leq\text{R})$

5.3 Characterizing Limits

Finally, we present some equations regarding joins and limits. Since limits are essentially (possibly-)infinitary joins, we write them using \bigvee .

$\bigvee : \forall \{c\} \rightarrow (El \ c \rightarrow \text{SMBTree}) \rightarrow \text{SMBTree}$
 $\bigvee f = \text{Lim } _ f$

Limits are an upper bound, so joining any element from a sequence with the limit of that sequence has no effect:

$\bigvee\text{-Bound} : \forall \{c : \mathbb{C}\} \{f : El \ c \rightarrow \text{SMBTree}\} \{k\}$
 $\rightarrow f \ k \vee (\bigvee f) \approx \bigvee f$
 $\bigvee\text{-Bound} = \text{ord} \rightarrow \text{equiv } (\leq\text{-limUpperBound } _)$

Moreover, a limit is an actual supremum: the limit of a function is absorbed by any upper bound of the function's image.

$\bigvee\text{-Supremum} : \forall \{c : \mathbb{C}\} \{f : El \ c \rightarrow \text{SMBTree}\} \{t\}$
 $\rightarrow (\forall k \rightarrow f \ k \vee t \approx t) \rightarrow (\bigvee f) \vee t \approx t$
 $\bigvee\text{-Supremum } \{f = f\} \text{lt}$
 $= \text{ord} \rightarrow \text{equiv } (\leq\text{-limLeast } (\lambda k \rightarrow \text{equiv} \rightarrow \text{ord } (\text{lt } k)))$

The join of a constant, non-empty sequence is the singular element of that sequence:

$\bigvee\text{-const} : \forall \{c \ t\}$
 $\rightarrow El \ c$
 $\rightarrow \text{Lim } c \ (\lambda _ \rightarrow t) \approx t$
 $\bigvee\text{-const } k = \leq\text{-limLeast } (\lambda _ \rightarrow \leq\text{-refl}) , \leq\text{-limUpperBound } k$

The join of an empty sequence is zero:

$\bigvee\text{-empty} : \forall \{c \ f\}$
 $\rightarrow \neg (El \ c)$
 $\rightarrow \text{Lim } c \ f \approx \mathbb{Z}$
 $\bigvee\text{-empty } \text{empty}$
 $= (\leq\text{-limLeast } (\lambda k \rightarrow \text{contradiction } k \ \text{empty})) , \leq\text{-Z}$

More interesting is the interaction between limits and joins. For two limits over the same index set, the join of those two limits is the same as the limit of joining the sequences together.

$\bigvee\text{-distHomo} : \forall \{c : \mathbb{C}\} \{f \ g : El \ c \rightarrow \text{SMBTree}\}$
 $\rightarrow (\text{Lim } c \ f) \vee (\text{Lim } c \ g) \approx \text{Lim } c \ (\lambda x \rightarrow f \ x \vee g \ x)$
 $\bigvee\text{-distHomo}$
 $= \text{max-LUB}$
 $(\leq\text{-extLim } (\lambda _ \rightarrow \text{max-}\leq\text{L}))$
 $(\leq\text{-extLim } (\lambda _ \rightarrow \text{max-}\leq\text{R}))$
 $, \leq\text{-limLeast}$
 $(\lambda k \rightarrow \text{max-mono}$
 $(\leq\text{-limUpperBound } _)$
 $(\leq\text{-limUpperBound } _))$

We can obtain a more general result, distributing a limit over a join, so long as the limit is over a non-empty sequence.

The join of a non-zero tree with an empty limit will be non-zero, but pushing the join under a limit will produce zero, so the following result only applies to non-empty limits.

$$\begin{aligned}
& \bigvee\text{-distHet} : \forall \{c : \mathbb{C}\} \{f : El\ c \rightarrow \text{SMBTree}\} \{t\} \\
& \rightarrow El\ c \\
& \rightarrow (\bigvee f) \vee t \approx \bigvee (\lambda x \rightarrow f\ x \vee t) \\
& \bigvee\text{-distHet}\ k \\
& = \text{max-LUB} \\
& \quad (\leq\text{-extLim}\ (\lambda _ \rightarrow \text{max-}\leq\text{L})) \\
& \quad (\leq\text{-limUpperBound}\ k \leq \leq\text{-extLim}\ (\lambda _ \rightarrow \text{max-}\leq\text{R})) \\
& \quad , \leq\text{-limLeast}\ (\lambda k \rightarrow \text{max-monol}\ (\leq\text{-limUpperBound}\ _))
\end{aligned}$$

6 Discussion

6.1 Comparison to Other Ordinal Systems

In the literature, many different variations of ordinals have been presented. To keep our comparison brief, we refer to the work of Kraus et al. [2023]. They give a comprehensive overview of ordinal notation systems in type theory, with a detailed comparison of their comparative properties. They define three different systems: Cantor normal forms that represent ordinals as binary trees, restricted Brouwer trees that represent ordinals as infinitely branching trees, and well-founded types that represent ordinals as types with a certain sort of relation on their elements.

The definitions Kraus et al. give are more restrictive than ours. For example, for Brouwer trees they require that `Lim` only operate on functions that are strictly increasing, preventing the definition of `limMax`. These restrictions make their ordinals very well-behaved with respect to propositional equality, so they can examine their mathematical properties. SMB-trees have less rich theory, but the properties they do satisfy are specifically tailored to proving termination of higher-order programs.

Transitivity, Extensionality and Well-foundedness Kraus et al. show three properties for each system they present: transitivity of the ordering, well-foundedness (as in Section 2.2), and *extensionality*, the property that two ordinals are equal iff their sets of smaller terms are equal. They also show a strict version of extensionality for each system: to ordinals are equal iff their sets of strictly smaller terms are equal.

SMB-trees satisfy each of the above properties: the transitivity of \leq is inherited from Brouwer trees, and we show well-foundedness in Section 2.2. Extensionality for \leq is trivially true for our setoid version of equivalence. For propositional equality, extensionality cannot be proved without some form of quotient type. We conjecture that the strict order $<$ is not extensional for SMB-trees, since it does not hold for Brouwer trees without quotient types.

Well-founded types lack a basic transitivity property for the strict order: without additional axioms, one cannot conclude $x < z$ from $x \leq y$ and $y < z$. So, though well-founded

types have binary and infinitary suprema like SMB-trees, they lack the basic principles for reasoning about strict orders, making them ill-suited for defining recursive procedures.

Classifiability Classifiability is the property that each ordinal is either zero, a successor, or a limit, and that exactly one of those properties holds. Restricted Brouwer trees and Cantor normal forms both satisfy classifiability, but SMB-trees do not. Even our version of Brouwer trees do not have this property: since we allow non-increasing sequences, the limit of the constant-zero sequence is equivalent to zero.

Not having classifiability does negatively affect the decidability properties of SMB-trees. For example, for restricted Brouwer trees, it is decidable whether a tree is infinite or not, but this is not the case for SMB-trees, since some limits are actually finite. However, since SMB-trees are defined specifically around well-founded recursion, losing decidability properties is an acceptable compromise. Additionally, the ability to reason about SMB-trees using the equational style reduces the need to pattern match on them.

Joins and Suprema The main novelty of SMB-trees is the existence of both binary suprema (joins) and infinitary suprema (limits) that interact well with the strict ordering. Cantor normal forms have binary joins and strict monotonicity (as a by-product of decidable ordering), but lack infinitary joins. Well-founded types have binary and infinitary suprema, but without additional axioms even their successor function is not monotone, so strict monotonicity is out of the question. For restricted Brouwer trees, binary joins cannot exist without further axioms. This is an artifact of allowing `Lim` only on strictly increasing sequences, since it disallows `limMax` or other similar constructs. So even without strict monotonicity, the capability of SMB-trees exceeds that of restricted Brouwer trees. The cost of this is that SMB-trees fulfill fewer nice properties with respect to propositional equality. Since setoid reasoning is sufficient for well-founded recursion, we find this tradeoff acceptable.

Conclusion Designing an ordinal library is an exercise in compromise, balancing the desired properties with the limitations of decidability and constructive reasoning. With SMB-trees, we have identified a point in the design space well suited to proving termination. The algebraic framework of SMB-trees lays the groundwork for future developments on reasoning mechanically about ordinals. Beyond of our specific use-case, the development of SMB-trees shows that sometimes careful design with dependent types can avoid the need for additional axioms or language features.

References

Agda-Developers. 2017. Github Issue: Equality is incompatible with sized types. <https://github.com/agda/agda/issues/2820>.

- Guillaume Allais, Edwin Brady, Nathan Corbyn, Ohad Kammar, and Jeremy Yallop. 2023. Frex: dependently-typed algebraic simplification. [arXiv:cs.PL/2306.15375](https://arxiv.org/abs/2306.15375)
- Yves Bertot and Pierre Castéran. 2004. Interactive Theorem Proving and Program Development. Springer-Verlag.
- Marc Bezem and Thierry Coquand. 2022. Loop-checking and the uniform word problem for join-semilattices with an inflationary endomorphism. *Theoretical Computer Science* 913 (2022), 1–7. <https://doi.org/10.1016/j.tcs.2022.01.017>
- Edwin C. Brady. 2021. Idris 2: Quantitative Type Theory in Practice. CoRR abs/2104.00480 (2021). [arXiv:2104.00480](https://arxiv.org/abs/2104.00480) <https://arxiv.org/abs/2104.00480>
- Jonathan H.W. Chan. 2022. Sized dependent types via extensional type theory. Master's thesis. University of British Columbia. <https://doi.org/10.14288/1.0416401>
- Alonzo Church. 1938. The constructive second number class. *Bull. Amer. Math. Soc.* 44, 4 (1938), 224 – 232.
- Nathan Corbyn. 2021. Proof Synthesis with Free Extensions in Intensional Type Theory. Technical Report. University of Cambridge. MEng Dissertation.
- Patrick Cousot and Radhia Cousot. 1979. Constructive versions of tarski's fixed point theorems. *Pacific J. Math.* 82, 1 (May 1979), 43–57. <https://doi.org/10.2140/pjm.1979.82.43>
- Leonardo de Moura, Soonho Kong, Jeremy Avigad, Floris van Doorn, and Jakob von Raumer. 2015. The Lean Theorem Prover (System Description). In *Automated Deduction - CADE-25*, Amy P. Felty and Aart Middeldorp (Eds.). Springer International Publishing, Cham, 378–388.
- Joey Eremondi. 2023a. JoeyEremondi/smb-trees: An Agda Library for Strictly Monotone Brouwer Trees. <https://doi.org/10.5281/zenodo.10204397>
- Joseph S. Eremondi. 2023b. On the design of a gradual dependently typed language for programming. Ph.D. Dissertation. University of British Columbia. <https://doi.org/10.14288/1.0428823>
- John Hughes, Lars Pareto, and Amr Sabry. 1996. Proving the Correctness of Reactive Systems Using Sized Types. In *Proceedings of the 23rd ACM SIGPLAN-SIGACT Symposium on Principles of Programming Languages (POPL '96)*. Association for Computing Machinery, New York, NY, USA, 410–423. <https://doi.org/10.1145/237721.240882>
- S. C. Kleene. 1938. On notation for ordinal numbers. *The Journal of Symbolic Logic* 3, 4 (1938), 150–155. <https://doi.org/10.2307/2267778>
- Nicolai Kraus, Fredrik Nordvall Forsberg, and Chuangjie Xu. 2023. Type-theoretic approaches to ordinals. *Theoretical Computer Science* 957 (2023), 113843. <https://doi.org/10.1016/j.tcs.2023.113843>
- Conor McBride and James McKinna. 2004. The view from the left. *Journal of Functional Programming* 14, 1 (2004), 69–111. <https://doi.org/10.1017/S0956796803004829>
- Ulf Norell. 2009. Dependently Typed Programming in Agda. In *Proceedings of the 4th International Workshop on Types in Language Design and Implementation (TLDI '09)*. ACM, New York, NY, USA, 1–2. <https://doi.org/10.1145/1481861.1481862>
- The Univalent Foundations Program. 2013. *Homotopy Type Theory: Univalent Foundations of Mathematics*. <https://homotopytypetheory.org/book>, Institute for Advanced Study.
- Andrea Vezzosi, Anders Mörtberg, and Andreas Abel. 2019. Cubical Agda: A Dependently Typed Programming Language with Univalence and Higher Inductive Types. *Proc. ACM Program. Lang.* 3, ICFP, Article 87 (jul 2019), 29 pages. <https://doi.org/10.1145/3341691>